
1. Introduction 

One of the best outcomes of the PLCopen specifications for Motion Control, Safety and Communication, 

(see www.PLCopen.org) is the definition provided for function block input and outputs. This provides a 

clear and concise shell as a starting point when considering the type of application level function to be 

created. For this two main function block categories are specified: the Execute and the Enable model. The 

goal now is to extend these to fit other application areas, and helping users to specify and implement 

consistent sets of FB libraries for their own usage. 

For this reason a working group was started within the PLCopen activity on Software Construction 

Guidelines defining the guidance for creating PLCopen compliant Function Block Libraries. 

 

By strictly following a few key features of the PLCopen specification, application level function blocks 

can provide a high degree of robustness, usability and predictability.  The behavior described makes it 

very easy to incorporate and debug functions in an application. Errors and ErrorIDs can be elevated to the 

calling functions. Interlocks are easier to create. Linking activities becomes easier. 

 

1.1. Naming conventions for this document 

In line with the PLCopen Coding Guidelines, the following naming conventions are used in this 

document: 

 

Prefix yes or no Used both in text (for readability) and in examples (when applicable) 

 

With prefix there is a difference between xError, eError and iError, compared to no prefixing with names 

like Error and ErrorID. It is decided to use unique names even without pre-fixing. Meaning that in this 

document xError and eErrorID are used in examples and Error and ErrorID can be used in the text. 

Also it is advised to use the same capitalization for every object instance, even if the tool/compiler doesn't 

mandate it. The following guidelines are proposed: 

 Use UPPER_SNAKE_CASE for CONSTANTS and user defined datatypes and keywords (like 

BOOL, FOR, TYPE and END_TYPE). 

 Use UpperCamelCase for all other multi-word items 

Variable names will be written in Courier New font size 11, while states will be written in the 

normal font in italic. 

 

1.2. Notes on the examples and usage of EN/ENO 

Any programs listed in this document are just examples, are not tested in practice and can be incorrect 

and not be useful for your application. This means that the code should just be seen as a learning basis 

and not as an implementation basis. 

Any code in this document should be considered as an example only. There is no need to implement the 

functionality exactly as proposed, just the interfaces of the function blocks and the state diagrams are 

necessary for compliance. 

The Enable /Execute constructs are on top of the usage of EN/ENO constructs (see Ch. 6.6.1.5. Execution 

Control (EN, ENO) of the IEC 61131-3 standard 3
rd

 edition). 

http://www.plcopen.org/

2. Commonalities in existing PLCopen specifications 

2.1. Motion Control, Safety and Communication 

This chapter is created to provide an overview of the current specification of PLCopen Function blocks 

over the different working groups: Motion, Safety and Communication. This is only done in short form, 

and for more specific explanation one is referred to the original specifications. 

The proposal for the generic function blocks is explained in the next chapters, especially Chapter 5 - 

Explanation of Rising Edge triggered FBs and Chapter 6 - Explanation of a Level Controlled FB. 

 

2.2. Function Block Models 

There are two basic function block models as described below. These functions are shown with the 

minimum inputs and outputs. Notice that Execute pairs with Done, and that Enable pairs with Valid. 

This aids the visual appeal of the code structure in FBD format when contacts and coils are connected to 

the function block. 

 

Programming to account for the variants will be described in detail in the following sections. All 

implementations will include added inputs and outputs for an actual application as shown in blue below. 

 

2.3. Motion Control: General structure 

PLCopen has created a suite of specifications for motion control. Within this suite an effort was made to 

have a consistent specification and layout of the function blocks. 

This chapter provides an overview of the commonalities in the specifications of the FBs. For more details 

check Appendix 4 Behaviour of inputs and outputs in PLCopen Motion Control FBs. 

 

The definition of the PLCopen FBs for motion control consists of an activation related section and a 

status related section. 

To activate an FB one originally had two options 

1. Execute as input, triggering the execution of the FB, and Done (or InVelocity, InGear, InTorque 

or InSync) as related output showing when the FB has finalized the command; 

2. Enable as input, level sensitive, and Valid as related output 

With the release of Version 2.0 of Part 1, the input ‘Continuous Update’ was added to combine both.  

 If it is TRUE, when the function block is triggered (rising ‘Execute’), it will - as long as it stays 

TRUE – make the function block use the current values of the input variables and apply it to the 

ongoing movement.  

 If ‘ContinuousUpdate’ is FALSE with the rising edge of the ‘Execute’ input, a change in the input 

parameters is ignored during the whole movement and the original behavior of previous versions 

is applicable. 

 

For the status the following outputs are defined: 

 Busy - The FB is not finished and new output values are to be expected 

 Active - Indicates that the FB has control of the axis 

 CommandAborted - ‘Command’ is aborted by another command 

 Error - Signals that an error has occurred within the Function Block 

 ErrorID - Error identification 

 

Example of a Motion Control Function Block 

As an example of a PLCopen Motion Control FB, a graphical representation of the FB 

MC_MoveAbsolute is shown here. 

Figure 1: The behavior of the ‘Execute’ / ‘Done’ in relevant FBs 

The behavior of the ‘Execute’ / ‘Inxx’ style FBs is as follows: 

Active

 
Figure 2: The behavior of the ‘Execute’ / ‘Inxx’ in relevant FBs 

2.4. Safety 

PLCopen has created a suite of specifications for safety. Also within this suite an effort was made to have 

a consistent specification and layout of the function blocks, in line with motion control. 

This chapter provides a short overview of the commonalities of the specifications of the safety FBs. 

Due to the safety related character of this part of the specification a reduction in the datatypes and 

functionalities is applicable, as well as the introduction of a Safe datatype. This means that there is a 

smaller overlap in outputs with the motion specification. Also the combination Execute/Done is 

replaced by Activate/Ready due to a slightly different behavior. 

 

Example of a Safety FB 

 

  SF_EmergencyStop   

BOOL  Activate  Ready  BOOL 

SAFEBOOL  S_EStopIn  S_EStopOut  SAFEBOOL 

SAFEBOOL  S_StartReset  Error  BOOL 

SAFEBOOL  S_AutoReset  DiagCode  WORD 

BOOL  Reset     

       

 

2.5. Specifications in Communication 

Together with the OPC Foundation, PLCopen has created FBs for the communication via OPC UA, 

Unified Architecture. This chapter gives a short overview of the specification. 

Also here the combination Execute/Done is applicable, as well as Busy, Error and ErrorID with 

functionalities equal to the PLCopen Motion Control specification. 

Example of a Communication FB 

 

2.6. Conclusion 

Overall there are 2 levels of commonality specified over the different PLCopen specifications: 

 

 Basic level with Execute/Done (or Activate/Ready) and Busy, Error and ErrorID, 

although the ErrorID can be an INTEGER, a WORD or a DWORD. 

 Extended Level with the addition of Active and CommandAborted  

Busy

Buffermode

Acceleration

Deceleration

Error

ErrorID

Figure 3: Common behaviour parameters 

Overview of related parameters in different libraries. 

Library MC v1.0 MC v2.0 Safety Communication 

 Enable Execute Enable/Execute   

Execute  X X  X 

ContinuousUpdate   X   

Enable X  X   

Activate    X  

Ready    X  

Valid X  X   

Enabled X     

Done  X X  X 

Busy X X X  X 

Active X X X   

CommandAborted  X X   

Error X X X X X 

ErrorID X X X  X 

DiagCode    X  

 

  UA_Connect   

BOOL  Execute  ConnectionHdl  DWORD 

STRING  ServerEndpointUrl   Done  BOOL 

STRUCT  SessionConnectInfo  Busy  BOOL 

TIME  Timeout   Error  BOOL 

    ErrorID  DWORD 

3. Introduction of the PLCopen Function Block concepts 

In this chapter the basic concepts of the PLCopen Function Blocks are explained. These concepts can also 

be called PLCopen Common Behaviour Model. There are 2 main groups identified: Edge Triggered and 

Level Controlled. The details are listed in Chapter 5 and 6 as well as in Appendix 1. 

3.1. Relation of Execute and Enable inputs to Level and Trigger inputs 

In the specifications of PLCopen FBs as described above, the inputs Enable and Execute are used to 

start the FB. However, further in this document a more generic description is used: 

 

 Edge Triggered Function Blocks – which are coupled to the Execute input; 

 Level Controlled Function Blocks – which are coupled to the Enable input. 

Sometimes it is important to choose a level-controlled model rather than an edge-triggered model. For the 

detection of a rising edge in a function block, two PLC cycles are necessary. Thus, if the requirement is to 

be able to process a new value in each cycle, an edge-triggered model cannot serve as a solution. In this 

case, a level-controlled function block model is the preferred way to implement the required functionality. 

3.2. Introduction to Edge Triggered function blocks 

As an example of an edge triggered function block the ETrigA (Edge Triggered with Abort functionality) 

is shown here, both the graphical representation and the state diagram. 

 

 

 
Edge Triggered function blocks in the context of this document are defined as follows: 

 The input variable xExecute is the defining feature for this type of function block. 

 A rising edge detected at the input variable xExecute (start condition) starts the operation 

defined by this particular edge triggered function block. 

 All inputs other than xExecute and the eventually present variable xAbort are sampled with this 

rising edge. These two inputs will be stored locally. Thus, later changes of these inputs cannot 

influence the defined operation while it is running [1]. 

 The input variable xExecute can be set to FALSE after the status TRUE was seen on the output 

variable xBusy. 

 A falling edge detected at the input variable xExecute will not abort the defined operation. The 

defined operation is running normally to its ready condition, abort condition or error condition. 

                                                 

[1] 
Sometimes it is necessary to have additional input variables which can influence the internal work flow. In this case, the 

special behavior of these variables should be clearly documented. 

[2] 
Sometimes it is necessary to have additional output variables with a valid status while xDone is not set to TRUE. In this 

case, the special behavior of these variables should be clearly documented. 

[3] 
Sometimes it is necessary to have additional output variables which a valid only in combination with the status of some 

other output variables. In this case, the special behavior and relationship of these variables should be clearly documented. 

 Only if the status TRUE is detected on the (eventually present) input variable xAbort the defined 

operation is cancelled (abort condition). 

 If the input variable xAbort is present and the input variable xExecute have the same value 

TRUE the abort condition is reached immediately. 

 Only one of the output variables xDone, xBusy, xError or the output variable xAborted (if 

present) can have the status TRUE at the same time. 

 If an abort condition was detected, the output variable xAborted is set to TRUE after setting the 

output variable xBusy to FALSE. 

 With the falling edge of xBusy the input variable xExecute is sampled and its inverted value is 

stored as a reset request inside the FB. 

 The state of the output variables will be valid for a minimum of one invocation even though the 

state of the xExecute input variable is already set to FALSE. In this case (reset request) the 

internal state of the function block is automatically reinitialized. In the other case (xExecute is 

still TRUE) the function block is waiting for a falling edge of the input variable xExecute before 

reinitializing the function block (standard handshake). 

 The status of the output variables other than xDone, xBusy, xError, xAborted or eErrorID 

are valid only while xDone has the status TRUE [2][3]. 

 With an active reset request and after the status TRUE of one of the output the variables xDone, 

xError or xAborted was seen, the input variable xExecute can be set to TRUE again and the 

function block will restart its defined operation (quick handshake). 

See the detailed descriptions of the reference implementation for the different edge triggered function 

blocks in the appendix: 

ETrig | ETrigTl | ETrigTo | ETrigTlTo | ETrigA | ETrigATl | ETrigATo | ETrigATo | ETrigATlTo 

 

3.3. Introduction to Level Controlled function blocks 

As an example of a level controlled function block the LCon (Level Controlled function block) is shown 

here, both the graphical representation and the state diagram. 

 

 

 
Level Controlled function blocks in the context of this document are defined as following: 

 The input variable xEnable is the defining feature for this type of function block. 

 The status TRUE detected on the input variable xEnable (start condition) starts the operation 

defined by this particular level controlled function block. The defined operation is running to its 

ready condition or error condition while the input variable xEnable is TRUE. The status FALSE 

detected on the input variable xEnable is interpreted as an abort (abort condition). This means 

the internal state of the function block and all output variables will be reinitialized and it is then 

ready for a new start (standard handshake). 

 The input variables will not be stored locally and can so influence the current work flow of the 

defined operation. 

 Only one of the output variables xDone, xBusy or xError can have the status TRUE at the same 

time. 

 The status of all output variables is valid as long as the status of the output variables xBusy or 

xDone is TRUE [3]. 

 With the falling edge of xBusy the input variable xEnable is sampled and its inverted value is 

stored as a reset request. 

 The state of the output variables will be valid for a minimum of one invocation even though the 

state of the xEnable input variable is already set to FALSE. In this case (reset request is TRUE) 

the internal state of the function block is automatically reinitialized. This can especially happen 

after an error condition while aborting the defined operation. 

 The status of the output variables other than xDone, xBusy, xError or eErrorID are valid only 

while xDone has the status TRUE [2][3]. 

 With an active reset request and after the status TRUE of one of the output variables xDone or 

xError was seen, the input variable xEnable can be set to TRUE again and the function block 

will restart its defined operation (quick handshake). 

Sometimes it is necessary to have a behavior model which never reaches its ready condition. An example 

is the MC_Power motion control function block. Also, the TCPServer function block as shown in Par. 3.7 

Cooperation of various function blocks is an example which needs a clear start condition but will never 

finishes the defined operation. 

Such a Level Controlled function block has no xDone output variable and no Done state. This behaviour 

is defined as a Continuous Behaviour.  

In the context of this document we refer to this kind of behavior models as LConC and LConTlC. 

See the detailed descriptions of the reference implementation for the different level controlled function 

blocks in the appendix (for LCon, LCOnTl, LConTo, LConTlTo, LConC and LCOnTlC). 

 

3.4. Common properties of these types of function blocks 

 If a specific invocation of a function block detects a start condition, the output variable xBusy is 

set immediately to the status TRUE. 

 As long as the defined operation of a function block is running, the output variable xBusy has the 

value TRUE. 

 If the ready condition is reached, the output variable xDone is set to TRUE and the output variable 

xBusy is set to FALSE. 

 If the error condition is reached, the output variable xError is set to TRUE and the output 

variable xBusy is set to FALSE. Additionally the output variable eErrorID will set to an error 

code other than ERROR.NO_ERROR. The eErrorID is defined here as an ENUM although users can 

also define this as INTEGER, WORD, DWORD or other datatype. 

 If the defined operation can be fully processed in one invocation, the ready condition or the error 

condition is reached immediately and the TRUE status of the output variable xBusy is never be 

seen. 

 With the rising edge of xDone, the status of all output variables will be frozen. 

 As long as one of the output variables xDone, xBusy or xError has the status TRUE the defined 

operation of this function block has not yet completed, so a further invocation is necessary. 

 

Timing constraints of these function blocks: 

 udiTimeLimit ([µs], 0 ⇒ no operating time limit): 

A function block could, for example, complete a complex task in a loop. The larger the task is, the 

more time that is consumed in the current invocation of this function block. The udiTimeLimit 

parameter can define how much time per invocation is permitted for consumption in the respective 

function block. Function blocks with an udiTimeLimit input variable are implemented in such 

a way that the current invocation is exited when the task is complete (Ready Condition), or when 

the consumed time for this invocation has exceeded the settings from udiTimeLimit. 

 

 udiTimeOut ([µs], 0 ⇒ no operating time limit): 

When processing its defined operation, a function block could be forced for example to wait for an 

external event. It can do this in an internal loop (Busy Wait) or it can check in each invocation 

whether its task can be completed in full. The udiTimeOut parameter can define then how much 

time is permitted for consumption for the defined operation. Function blocks with the 

udiTimeOut input variable are implemented in such a way that the current invocation is exited 

towards an error condition (xError ⇒ TRUE and eErrorID ⇒ ERROR.TIME_OUT) when the 

time interval as defined by udiTimeOut has been exceeded. 

 

3.5. Error Domains and Error Codes 

Every library provides its own error domain (Error Domain = Library namespace). 

Every library provides a set range of possible values for an ErrorID output variable (Error Codes = 

ERROR enum data type). 

All Function Blocks in this document have a Boolean output xError to indicate that an error condition 

has been reached. In that case the related information will be signaled with the value of the output 

eErrorID. The eErrorID represents an Integer, indicating the reason of the error. In many cases this 

integer value is used as input for an additional FB which converts the number to a related localized string 

in an applicable language. The set of values for a specific eErrorID are application dependent. In case 

several libraries are combined (several domains), there can be an overlap in the numbers of the 

eErrorID, meaning that the same number identifies a different error in a different domain. For this 

reason a value range definition for eErrorID per library must be done. 

The error handling of a function block should be designed in a way that only error codes are returned, 

which are documented in the affected library. It is very convenient but not recommended simply to return 

untreated error codes from sub libraries. This would result in a bad user experience. It is recommended to 

map foreign error codes to the error range of the affected library. 

In the following example we take a closer look to the relationship between two libraries, each with a 

specific domain of error codes. The first library may be called the “Memory Block Manager library” and 

is built in the namespace MBM. The second library may be called the “Function Block Factory” and is built 

in the namespace FBF. Each library defines its own ERROR enum data type. 

 Two libraries are isolated with a namespace (in this example FBF and MBM). 

 Each ERROR Enum declaration should respect two predefined error codes. 

o NO_ERROR ⇒ 0 (Zero) 

o TIME_OUT ⇒ 1 (One) 

 If the TIME_OUT error code has no usage in a specific domain the value should not be reused for 

another error code. 

 Any error code needs a short description about the background of its error condition. 

 An enum data type should be isolated from other enum data types with its own namespace 

({attribute 'qualified_only'}). FBF.ERROR.NO_MEMORY has a completely different 

meaning as MBM.ERROR.NO_MEMORY. 

Working together with sub libraries brings up the need for mapping the different error domains to the one 

local domain. The next example demonstrates the possible design of an error code mapping function. It 

handles the error codes (from CS.ERROR and CO.ERROR) of two sub libraries and tries to map these to 

the one local Error Enum (CANOPEN_KERNEL_ERROR) (All enum data types in this example have the 

base type INT). 

 
 

This design assumes CS.ERROR.NO_ERROR has the same value as CO.ERROR.NO_ERROR and the rest of 

the value range of CS.ERROR and CO.ERROR is disjunct. 

3.6. How to handle the STATE enum data type 

The example implementations of the different behaviour models in this document depend strongly on the 

processing of a central state machine and its handling of the STATE enum data type. For each 

implementation there is a specific simplified picture of the STATE enum data type included, which is 

purely for readability and ease of comprehension. In a real library which may be implementing all the 

different behaviour models as specific function blocks we will find only one STATE enum data type. This 

data type will define all possible states in relation to all kinds of function blocks in the PLCopen 

behaviour model function block family. 

 

 
 

3.7. Cooperation of various function blocks 

The cooperation of various function blocks each follow one of the previous described behaviour models. 

Here an example is shown of this cooperation. 

 

If the design of function block interfaces consistently takes into account this PLCopen Common 

Behaviour Model then complex relationships between instances of them can easily be expressed and 

understood, in particular in graphical languages. This example demonstrates the opportunities which are 

resulting from the use of the PLCopen Common Behaviour Model. 

TCP Reader Example

 

A TCP Server(LConC) listens at a specific endpoint. This endpoint is defined with a tuple consisting of 

an IP address and a related port number. After a connection to a TCP client has been established, the 

related Connection (LConC) will be activated. With the help of a Reader (LConTlC), a message 

structure is allocated out of a MessagePool. This structure is then used to store the information received 

from the client. For further processing, the message is then handed over to function block QWrite(LCon) 
that interfaces a queue. 

Extending the Example to a Complete EchoServer 

 

A Writer (ETrigTlTo) is working now at the end of the original Queue and is requesting (dequeuing) 

the stored messages. This message is sent back over the original Connection to the client. So every 

received message will be sent back to the original sender, the client. With this small modification, a 

complete Echo Server is created. 

Note: 

The Reader function block in the example above must have the possibility to provide a new 

message in each cycle. If there is no new message available in the current cycle, the output variable 

is set to null. So in this case, a level-control model was chosen. 

Sending a message with the help of the Writer function block in the example above can last more 

than one cycle. So it is important to choose an edge-triggered model. 

In order to combine these two function blocks into an efficient working team, a queue mechanism 

is necessary. 

Note the connection between the xBusy output variable of the Writer and the negated xExecute 
input variable of the QReader as handshake: in this way no extra cycle is necessary and no data will 

be lost, making it very efficient. 

 

Transformation to a Multithreaded EchoServer 

Step 1: Encapsulate the “Reader – Queue – Writer” logic into a function block called EchoHandler

 
 

Content of the new EchoHandler function block:

 

Step 2: Declare a number of EchoHandler instances and assign these instances to different tasks. 

The result is a complete multi-threaded EchoServer. 

4. Introduction in the object oriented features of IEC 61131-3 

The 3
rd

 Edition of the IEC 61131-3 introduces object oriented features to the standard. In particular, the 

standard defines methods, classes, interfaces, inheritance, etc. In this document those features are not 

further explained in detail. Those concepts are considered to be known to the reader. 

 

This document makes careful use of those features. Only protected and private methods are used to 

structure the code of the resulting function blocks.  

 

These function blocks define a set of state machines, that are considered to be very common in their 

behaviour. These function blocks are not supposed to be used directly in an application, but as base 

function blocks for further usage. To make use of such a function block, a new function block could be 

derived from the appropriate base function block by inheritance. The protected methods should be 

overridden to implement the specific behaviour. 

 

However, no feature of the 3
rd

 Edition is really necessary to create compliant function blocks. Every 

method call in the code of this document can be replaced by the code of the method. The base function 

blocks can then be considered as templates. To make use of the base function block, a new function block 

could be created by copying the base function block. The calls to the protected methods should be 

replaced to implement the specific behaviour. 

The basic FBs ETrig and LCon can be extended via inheritance to the FBs ETrigA, ETrigATl, ETrigATo, 

ETRigATlTo, ETrigTlTo, ETRigTo, ETRigTl and LConTl, LConTo, LConTlTo respectively. 

5. Explanation of Rising Edge triggered FBs 

5.1. The basic FB: ETrig 

This is the functionality in its most simple form with only an Execute as input, both in textual as well as 

in graphical representation.  

 

Textual representation Graphical representation 
FUNCTION_BLOCK ETrig 

VAR_INPUT 

    xExecute: BOOL; 

END_VAR 

VAR_OUTPUT 

    xDone: BOOL; 

    xBusy: BOOL; 

    xError: BOOL; 

    eErrorID: INT; 

END_VAR 
 

 

To describe the functionality and behaviour of this FB one can make use of a state diagram, describing 

the different states as well as the transitions as a result of an activity. 

Basically there are 4 states: Dormant, Busy and Done, combined with Reset and Error. See hereunder. 

Listed are also the transitions including error behaviours’. 

 

 
 

Figure 4: Basic State Diagram 

After a rising edge was detected at the input xExecute the internal state is switched from Dormant to 

Busy while in that invocation all the inputs are sampled and stored. The output xBusy will be set to TRUE. 

The defined operation will be started.  

While working on the defined operation, a number of conditions can appear that lead to the exit from the 

Busy state. This means the value of the output variable xBusy will be set to FALSE and the internal state 

will be switched from Busy to one of the states Done or Error. This change will be mirrored to one of the 

output variables xDone or xError. Only one output variable of this set of variables can have the status 

TRUE at the same time. 

Ready Condition: If the operation has reached its ready condition without any error and timing 

constraints the output variable xDone is set to TRUE. This means the internal state is switched 

from Busy to Done. 

Error Condition: If an error condition was detected, the output variable xError is set to TRUE. 

This means the internal state is switched from Busy to Error. In addition, one of the defined error 

codes (one value out of the local enumeration type ERROR) will be assigned to the output variable 

eError. 

The values TRUE of one of the output variables xDone or xError has to be stable for minimum one 

cycle. 

After a FALSE status for the input variable xExecute is detected, the internal state will be switched to 

Reset. 

All Outputs will be initialized to their default status (Reset Outputs). All claimed resources will be freed. 

Especially the output variables xDone and xError will be set to FALSE. 

After doing this reinitialization work, the internal state will be switched from Reset to Dormant. 

(Note: One has to make sure that the state TRUE of one of the output variables xDone or xError must be 

stable for a minimum of one cycle, e.g. add a sub-state Sync.) 

 

Example of the ST Program for the FB ETrig with OO 

For this example the Object Oriented features are used. For an example in the classical approach refer to 

Appendix 2 Example without using Object Oriented features. 

The following methods are used in the code: prvResetOutputs(), prvStart() and prvCyclicAction() 

 

The STATE Enumeration: 
 

            TYPE STATE : 

            ( 

                DORMANT,   // Waiting for Start 

                EXECUTING, // CyclicAction is running 

                DONE,      // CyclicAction is complete 

                ERROR,     // Error condition reached 

                RESETTING  // ResetAction is running 

            ); 

            END_TYPE 

 

The ERROR Enumeration: 
 

            TYPE ERROR : 

            ( 

                NO_ERROR :=0, 

                TIME_OUT := 1 

                (*...*) 

            ); 

            END_TYPE 

 

Implementation of the Function Block ETrig: 
 

            FUNCTION_BLOCK ETrig 

            VAR_INPUT 

                // Rising edge starts defined operation 

                // FALSE ⇒ reset the defined operation 
                // after ready condition was reached 

                xExecute: BOOL; 

            END_VAR 

 

            VAR_OUTPUT 

                // ready condition reached 

                xDone: BOOL; 

                // operation is running 

                xBusy: BOOL; 

                // error condition reached 

                xError: BOOL; 

                // error code describing error condition 

                eErrorID : ERROR; 

            END_VAR 

 

            VAR 

                 eState : STATE; 

                 xfIRSTiNVOCATION : BOOL := TRUE; 

                 xResetRequest : BOOL; 

            END_VAR 

 

            VAR_TEMP 

                 xAgain : BOOL; 

            END_VAR 

 

            REPEAT 

                xAgain := FALSE; 

                CASE eState OF 

                    STATE.DORMANT:   HandleDormantState(xAgain=>xAgain); 

                    STATE.EXECUTING:   HandleStartState(xAgain=>xAgain); 

                    STATE.DONE:    HandleDoneState(xAgain=>xAgain); 

                    STATE.ERROR:   HandleErrorState(xAgain=>xAgain); 

                    STATE.RESETTING:   HandleResettingState(xAgain=>xAgain); 

                END_CASE 

 

            UNTIL NOT xAgain 

            END_REPEAT; 

 

The Handler for the Dormant State 
 

METHOD PRIVATE FINAL HandleDormantState 

VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

 

IF xExecute THEN 

    xBusy := TRUE; 

    eState := STATE.EXECUTING; 

    xAgain := TRUE; 

END_IF 

 

The Handler for the Executing State 
 

METHOD PRIVATE FINAL HandleExecutingState 

VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

VAR 

    xComplete : BOOL; 

END_VAR 

 

CyclicAction( 

    xComplete=>xComplete, 

    eErrorID=>eErrorID 

); 

 

IF eErrorID <> ERROR.NO_ERROR THEN 

    eState := STATE.ERROR; 

    xAgain := TRUE; 

ELSIF xComplete THEN 

    eState := STATE.DONE; 

    xAgain := TRUE; 

END_IF 

 

The Handler for the Done State 
 

METHOD PRIVATE FINAL HandleDoneState 

VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

 

IF xDone AND (xResetRequest OR NOT xExecute) THEN 

    eState := STATE.RESETTING; 

    xAgain := TRUE; 

ELSE 

    xBusy := FALSE; 

    xDone := TRUE; 

    xResetRequest := NOT xExecute; 

    xAgain := FALSE; (* !!! *) 

END_IF 

 

The Handler for the Error State 
 

METHOD PRIVATE FINAL HandleErrorState 

VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

 

IF xError AND (xResetRequest OR NOT xExecute) THEN 

    eState := STATE.RESETTING; 

    xAgain := TRUE; 

ELSE 

    xBusy := FALSE; 

    xError := TRUE; 

    xResetRequest := NOT xExecute; 

    xAgain := FALSE; (* !!! *) 

END_IF 

 

The Handler of the Resetting State 
 

METHOD PRIVATE FINAL HandleResettingState 

VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

VAR 

    xComplete : BOOL; 

END_VAR 

 

ResetAction(xComplete=>xComplete); 

 

IF xComplete THEN 

    xBusy := FALSE; 

    xDone := FALSE; 

    xError := FALSE; 

    eErrorID := ERROR.NO_ERROR; 

    eState := STATE.DORMANT; 

    xFirstInvocation := TRUE; 

    xAgain := xResetRequest; (* !!! *) 

    xResetRequest := FALSE; 

END_IF 

 

Example implementation of the application specific Methods 

This code is listed here just as an example. The content needs to be adapted to the real requirements of a 

specific application. 

 

The Implementation of the CyclicAction 
METHOD PROTECTED CyclicAction 

VAR_OUTPUT 

    xComplete : BOOL; 

    eErrorID : ERROR; 

END_VAR 

 

IF xFirstInvocation THEN 

    (* Starting *) 

    // for the first (!) invocation, 

    // sample the input variables 

    xFirstInvocation := FALSE; 

END_IF 

 

(* Executing *) 

// working to reach the ready condition 

// ⇒ xComplete := TRUE 
// if an error condition is reached 

// ⇒ set eErrorID to a value other than ERROR.NO_ERROR 
 

xComplete := TRUE; 

eErrorID := ERROR.NO_ERROR; 

 

IF xComplete OR eErrorID <> ERROR.NO_ERROR THEN 

    (* Cleaning *) 

    // if possible free as much allocated resources 

    // as possible 

END_IF 

 

The Implementation of the ResetAction 
 

METHOD PROTECTED ResetAction 

VAR_OUTPUT 

    xComplete : BOOL; 

END_VAR 

 

// free all allocated resources 

// reinitialize instance variables 

 

xComplete := TRUE; 

5.2. Adding the Aborting Functionality to the basis 

The next step is to add the aborting functionality to the basic function block. With aborting the on-going 

functionality is interrupted with another command. For this we need an input (xAbort), to initiate the 

aborting functionality, and an output (xAborted) to reflect the status: it is SET when the FB is aborted.  

(Note that this representation is different from the PLCopen Motion Control Specification where the 

aborting functionality is hidden while shown here as an input) 

 

Textual representation Graphical representation 
FUNCTION_BLOCK ETrigA 

VAR_INPUT 

    xExecute: BOOL; 

    xAbort: BOOL; 

END_VAR 

VAR_OUTPUT 

    xDone: BOOL; 

    xBusy: BOOL; 

    xError: BOOL; 

    xAborted: BOOL; 

    eErrorID: INT 

END_VAR 

  
 

 

 

 

 
 

 

 

 

With this added functionality, the state diagram gets more complex since there is a state Abort added: 

 

 
Figure 5: The State Diagram of ETrigA 

Figure 6: State Diagram of ETrigA with output status 

Note that the Resetting State in this picture is included in the Dormant state. 

 

After a rising edge was detected at the input xExecute the internal state is switched from Dormant to 

Executing while in that invocation all the inputs are sampled and stored. 

The output xBusy will be set to TRUE. This means the internal state is switched to Busy. 

The defined operation will be started (CyclicAction). 

While working on the defined operation a number of conditions can appear, that lead to the exit from the 

Busy state. This means the value of the output variable xBusy will be set to FALSE and the internal state 

will be switched from Busy to one of the states Done, Error or Aborted. This change will be mirrored to 

one of the output variables xDone, xError or xAborted. Only one output variable of this set of 

variables can have the status TRUE at the same time. 

Ready Condition: If the operation has reached its ready condition without any error and timing 

constraints the output variable xDone is set to TRUE. This means the internal state is switched from 

Busy to Done. 

Error Condition: If an error condition was detected, the output variable xError is set to TRUE. 

This means the internal state is switched from Busy to Error. In addition, one of the defined error 

codes (one value out of the local enumeration type ERROR) will be assigned to the output variable 

eErrorID. 

Abort Condition: If a status of TRUE was detected for the xAbort input variable, the abort 

condition is reached. This means the internal state is switched from Busy to Abort. Any current 

action of the defined operation will be aborted. After this is done the output variable Aborted will 

be set to TRUE and the internal state is switched from Abort to Aborted. 

The value of TRUE of one of the output variables xDone, xError or xAborted has to be stable for 

minimum one cycle. 

After a FALSE status for the input variable xExecute is detected, the internal state will be switched to 

Reset. 

All Outputs will be initialized to their default status (Reset Outputs). All claimed resources will be freed. 

Especially the output variables xDone, xError and xAborted will be set to FALSE. 

After doing this reinitialization work, the internal state will be switched from Reset to Dormant. 

Example of a SFC Program 

Hereunder an example of the implementation in Sequential Function Chart, SFC of the state diagram of 

Figure 5: The State Diagram of ETrigA. The different states are clearly identifiable in the Steps, linked to 

the related Action Blocks with the Action Qualifiers. The transition conditions between the Steps are 

linked to the transitions in the State Diagram. 

 
Figure 7: Example of the ETrigA State Diagram in SFC 

Note to the used Action qualifiers: 

N Non-stored, executes while associated step is active 

R Resets a stored action 

S Sets an action active, i.e. stored 

For more details refer to the IEC 61131-3 standard. 

 

For an example of the code and the timing diagrams, refer to Appendix 1 Datasheets of the Edge 

Triggered and Level Controlled FBs Par. Appendix 1.3.5 ETrigA. 

5.3. Adding timer functionality 

On top of the abort functionality or without this functionality one can add timers to make the functionality 

more robust. Basically there are 3 options for timers: 

 

1. TimeOut (To): the overall operating time of the defined operation should be lower than the time 

(in µs) as specified by the input value udiTimeOut; 

2. TimeLimit (Tl): here the time limit is set that the operation stays within the cycle time. In that 

way a longer operation can be divided over several cycles; 

3. And the combination of them both (TlTo). 

 

What does udiTimeLimit do? 

A function block could, for example, complete a complex task in a loop. The larger the task is, the more 

time that is consumed in the current task of this function block. The udiTimeLimit parameter can 

define how much time per invocation is permitted for consumption in the respective function block. 

Function blocks with udiTimeLimit input must implement their CyclicAction in such a way that this 

method is exited when the task is complete (ReadyCondition, xComplete:=TRUE) or when the 

consumed time in this cycle has exceeded the settings from udiTimeLimit (xComplete:=FALSE).  

  

What does udiTimeOut do? 

When processing its cycle action, a function block for example could be forced to wait for an external 

event. It can do this in an internal loop (BusyWait) or it can check in each cycle whether its task can be 

completed in full. The udiTimeOut parameter can define then how much time is permitted for 

consumption in the Busy state.  

Function blocks with the udiTimeOut input must implement their CyclicAction in such a way that this 

method is exited towards xError (eError := ERROR.TIME_OUT) when the time interval as defined by 

udiTimeOut has been exceeded. 

 

Examples with timers without Aborting 

 
Note: the inputs and outputs are listed here without prefixes in the names 

 

Examples with Aborting and timers 

 
Note: the inputs and outputs are listed here without prefixes in the names 

 

For the state diagram let us look at the most extended version: ETrigATlTo. The condition TimeOut 

results in a transition to the state Error. The condition TimeLimit is applicable in Executing. 

Figure 8: State Diagram for ETrigATlTo 

Note: there are no R_Trig or F_Trig in this state diagram. The values of the parameters are relevant here. For 

example from Dormant to Executing one checks the value of xExecute 

 

Detailed description of the Function Block ETrigATlTo 

1. The function block is called inside of a POU one time per task cycle without any conditions. This 

is called an invocation. 

2. After a rising edge has been detected at the input variable xExecute, the internal state is switched 

from Dormant to Executing. 

3. The status of all other input variables then xExecute and xAbort will be sampled and stored 

locally (see xFirstInvocation inside CyclicAction). Thus, later changes of these inputs 

cannot influence the defined operation while it is running 0. 

4. The output xBusy will be set to TRUE. 

At this point in time it would be possible to set the input variable xExecute to the status FALSE 

(quick handshake). 

5. The defined operation will be started (see the comments inside CyclicAction). 

6. If the operating time for the current invocation is higher than the time (in µs) as specified by the 

input value xTimeLimit, the operation will be interrupted and continued in the next invocation 

(see xTimeLimit in CyclicAction). 

7. While working on the defined operation, a number of conditions can appear that lead to the exit 

from the Executing state. This means the value of the output variable xBusy will be set to FALSE 

and the internal state will be switched from Executing to one of the states Done, Error or Aborted. 

This change will be mirrored to one of the output variables xDone, xError or xAborted. Only 

one output variable of this set of variables can have the status TRUE at the same time. With the 

falling edge of xBusy the input variable xExecute is sampled and its inverted value is stored as a 

reset request (see xResetRequest in the methods HandleDoneState, HandleErrorState 
and HandleAbortedState). 

a. Ready Condition: 

If the operation has reached its ready condition without any error and timing constraints 

(see xComplete in CyclicAction), the output variable xDone is set to TRUE. This 

means the internal state is switched from Executing to Done. 

b. Error Condition: 

If an error condition was detected (see eErrorID in in CyclicAction), the output 

variable xError is set to TRUE. This means the internal state is switched from 

Executing to Error. In addition, one of the defined error codes (one value out of the 

local enumeration type ERROR) will be assigned to the output variable eErrorID. 

c. Timeout Condition: 

If the overall operating time of the defined operation is higher than the time (in µs) as 

specified by the input value udiTimeOut, a timeout condition will detected causing the 

output variable xError to be set to TRUE. This means the internal state is switched from 

Executing to Error. Furthermore, the output variable eErrorID is set to a special error 

code (ERROR.TIME_OUT). 

d. Abort Condition: 

If a status of TRUE was detected for the xAbort input variable, the abort condition is 

reached. This means the internal state is switched from Executing to Aborting. Any current 

action of the defined operation will be aborted. After this is done (see xComplete inside 

AbortAction), the output variable xAborted will be set to TRUE and the internal state is 

switched from Aborting to Aborted. If an error condition was detected (see eErrorID in 

in AbortAction), the output variable xError is set to TRUE. This means the internal 

state is switched from Aborting to Error. 

8. As a reaction to the rising edge of one of the output variables xDone, xError or xAborted it 
would be possible to set the status of the input variable xExecute again to TRUE (quick 

handshake). 

9. The value of TRUE of one of the output variables xDone, xError or xAborted must be stable for 

a minimum of one invocation. This means the internal states Done, Error or Aborted must be 

active for a minimum of one invocation. This property guarantees that the values of the output 

variables are valid and stable for a minimum of one invocation. 

10. The status of output variables other than xDone, xBusy, xError, xAborted or eErrorID are 

valid only while xDone has the status TRUE. 

11. After a FALSE status for the input variable xExecute is detected (standard handshake) or a reset 

request is active (quick handshake), the internal state will be switched from Done, Error or 

Aborted to Resetting. 

12. All outputs will be initialized to their default statuses (ResetAction). All claimed resources will 

be freed. Specifically, the output variables xDone, xError and xAborted will be set to FALSE. 

After executing the code of the ResetAction the function block should be prepared for a new 

switch of the internal state from Dormant to Executing. 

13. After doing this reinitialization work, the internal state will be switched from Resetting to 

Dormant (see xComplete inside ResetAction). 

 

5.4. Example of the ST Program for ETrigATlTo 

An example of a possible ST program is shown in Appendix 1.3.8 ETrigATlTo. 

6. Explanation of a Level Controlled FB 

6.1. Basic Level Controlled FB 

This is the equivalent of the previous most simple FB, but now as level controlled. 

 

Textual representation Graphical representation 
FUNCTION_BLOCK LCon 

VAR_INPUT 

    xEnable: BOOL; 

END_VAR 

VAR_OUTPUT 

    xDone: BOOL; 

    xBusy: BOOL; 

    xError: BOOL; 

    eErrorID: INT; 

END_VAR 

 

 

 

State Diagram Basic Level Controlled FB 

This state diagram also has 5 states: Dormant, Executing, Done, Error and Aborting, although for clarity 

we added one state in Figure 9: Overview State Diagram Level Controlled FB to show clearly the 

transition back to the state Resetting. In Figure 10: Example of the state diagram for a basic level 

controlled FB this state is again incorporated in the Dormant state. 

 
Figure 9: Overview State Diagram Level Controlled FB (LCon) 

 
Figure 10: Example of the state diagram for a basic level controlled FB (LCon) 

6.2. Example of the ST Program for the Function Block LCon in OO 

The STATE Enumeration: 

 
TYPE STATE : 

( 

    DORMANT,   // Waiting for xEnable 

    EXECUTING, // CyclicAction is running 

    ABORTING,  // AbortAction is running 

    DONE,      // Ready condition reached 

    ERROR,     // Error condition reached 

    RESETTING  // ResetAction is running 

); 

END_TYPE 

 

The ERROR Enumeration 
TYPE ERROR : 

( 

    NO_ERROR := 0, 

    TIME_OUT := 1 

    (* ... *) 

); 

END_TYPE 

 

Implementation of the Function Block LCon: 

 
FUNCTION_BLOCK LCon 

VAR_INPUT 

    // TRUE ⇒ activate the defined operation 
    // FALSE ⇒ abort/reset the defined operation 
    xEnable: BOOL;           

END_VAR 

VAR_OUTPUT 

    // ready condition reached 

    xDone: BOOL; 

    // operation is running 

    xBusy: BOOL; 

    // error condition reached 

    xError: BOOL; 

    // error code describing error condition 

    eErrorID : ERROR; 

END_VAR 

VAR 

    eState : STATE; 

    xResetRequest : BOOL; 

END_VAR 

VAR_TEMP 

    xAgain :BOOL; 

END_VAR 

 

REPEAT 

    xAgain := FALSE; 

    CASE eState OF 

        STATE.DORMANT: HandleDormantState(xAgain=>xAgain); 

        STATE.EXECUTING: HandleExecutingState(xAgain=>xAgain); 

        STATE.DONE: HandleDoneState(xAgain=>xAgain); 

        STATE.ERROR: HandleErrorState(xAgain=>xAgain); 

        STATE.ABORTING: HandleAbortingState(xAgain=>xAgain); 

        STATE.RESETTING: HandleResettingState(xAgain=>xAgain); 

    END_CASE 

UNTIL NOT xAgain 

END_REPEAT; 

 

 

The Handler for the Dormant State: 

METHOD PRIVATE FINAL HandleDormantState 

VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

 

IF xExecute THEN 

    xBusy := TRUE; 

    eState := STATE.EXECUTING; 

    xAgain := TRUE; 

END_IF 

 

The Handler for the Executing State: 

 
METHOD PRIVATE FINAL HandleExecutingState 
VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

VAR 

    xComplete : BOOL; 

    xTimeOut : BOOL; 

END_VAR 

       

IF xEnable THEN 

    CyclicAction( 

        xComplete=>xComplete, 

        eErrorID=>eErrorID 

    ); 

END_IF 

     

IF eErrorID <> ERROR.NO_ERROR THEN 

    eState := STATE.ERROR; 

    xAgain := TRUE; 

ELSIF NOT xEnable THEN 

    eState := STATE.ABORTING; 

    xAgain := TRUE; 

ELSIF xComplete THEN 

    eState := STATE.DONE; 

    xAgain := TRUE; 

END_IF 

 

The Handler for the Aborting State: 

 
METHOD PRIVATE FINAL HandleAbortingState 

VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

VAR 

    xComplete : BOOL; 

END_VAR 

 

AbortAction( 

    xComplete=>xComplete, 

    eErrorID=>eErrorID 

); 

 

IF eErrorID <> ERROR.NO_ERROR THEN 

    eState := STATE.ERROR; 

    xAgain := TRUE; 

ELSIF xComplete THEN 

    eState := STATE.RESETTING; 

    xAgain := TRUE; 

END_IF 

 

The Handler for the Done State: 

 
METHOD PRIVATE FINAL HandleDoneState 

VAR_OUTPUT 



PLCopen
®

 
for efficiency in automation 

 

Creating PLCopen Compliant Libraries  V 1.0 - Official Release © PLCopen (2017)  

Software Creation Guidelines May 4, 2017  page 34/89 

    xAgain : BOOL; 

END_VAR 

 

IF xDone AND NOT xEnable THEN 

    eState := STATE.RESETTING; 

    xAgain := TRUE; 

ELSE 

    xBusy := FALSE; 

    xDone := TRUE; 

    xAgain := FALSE; (* !!! *) 

END_IF 

 

The Handler for the Error State: 

 
METHOD PRIVATE FINAL HandleErrorState 

VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

 

IF xError AND (xResetRequest OR NOT xEnable) THEN 

    eState := STATE.RESETTING; 

    xAgain := TRUE; 

ELSE 

    xBusy := FALSE; 

    xError := TRUE; 

    xResetRequest := NOT xEnable; 

    xAgain := FALSE; (* !!! *) 

END_IF 

 

The Handler of the Resetting State: 

 
METHOD PRIVATE FINAL HandleResettingState 

VAR_OUTPUT 

    xAgain : BOOL; 

END_VAR 

VAR 

    xComplete : BOOL; 

END_VAR 

 

ResetAction(xComplete=>xComplete); 

 

IF xComplete THEN 

    xBusy := FALSE; 

    xDone := FALSE; 

    xError := FALSE; 

    eErrorID := ERROR.NO_ERROR; 

    eState := STATE.DORMANT; 

    xAgain := xResetRequest; (* !!! *) 

    xResetRequest := FALSE; 

END_IF 

 

 

Exemplary Implementation of the Application specific Methods 
This code is listed here just as an example. The content needs to be adapted to the real requirements of a 

specific application. 

 

The Implementation of the CyclicAction: 
METHOD PROTECTED CyclicAction 

VAR_OUTPUT 

    xComplete : BOOL; 

    eErrorID : ERROR; 

END_VAR 

 

IF xEnable THEN 

    (* Executing *) 

    // for every invocation, 

    // sample the input variables 

 

    // working to reach the ready condition 

    // ⇒ xComplete := TRUE 
    // if an error condition is reached set 

    // eErrorID to a value other than ERROR.NO_ERROR 

 

    xComplete := TRUE; 

    eErrorID := ERROR.NO_ERROR; 

END_IF 

 

IF NOT xEnable OR xComplete OR eErrorID <> ERROR.NO_ERROR THEN 

    (* Cleaning *) 

    // if possible free as much allocated resources 

    // as possible 

END_IF 

 

The Implementation of the AbortAction: 
  

METHOD PROTECTED AbortAction 

VAR_OUTPUT 

    xComplete : BOOL; 

    eErrorID : ERROR; 

END_VAR 

 

// abort all running operations 

// if an error condition is reached set 

// eErrorID to a value other than ERROR.NO_ERROR 

 

xComplete := TRUE; 

eErrorID := ERROR.NO_ERROR; 

 

The Implementation of the ResetAction: 

 
METHOD PROTECTED ResetAction 

VAR_OUTPUT 

    xComplete : BOOL; 

END_VAR 

 

// free all allocated resources 

// reinitialize instance variables 

 

xComplete := TRUE; 

6.3. Adding Timers 

Also here one can add timer functionalities: 

 

1. TimeOut (To): the overall operating time of the defined operation should be lower than the time 

(in µs) as specified by the input value udiTimeOut; 

2. TimeLimit (Tl) (and TlC – TimeLimit without Done output): here the time limit is set that the 

operation stays within the cycle time. In that way a longer operation can be divided over several 

cycles; 

3. And the combination of them both (TlTo) 

 

 
Figure 11: Representation of the added Timers to the level controlled FB 

Note: the inputs and outputs are listed here without prefixes in the names 

 

Example of LConTl 

 

 
Figure 12: State diagram LConTl 

Figure 13: Example of implementation of the state diagram LConTl in SFC 

The implementation of LConTl in Sequential Function Chart, SFC, is straightforward. The different states 

are clearly identifiable in the Steps, linked to the related Action Blocks with the Action Qualifiers. The 

transition conditions between the Steps are linked to the transitions in the State Diagram. 

 

State Diagram LConTlTo 

An example of the state diagram for LConTlTo, so a level controlled FB with time limit and time out, is 

shown hereunder: 

 
Figure 14: State Diagram for LConTlTo 

Detailed description of the Function Block LConTlTo 

1. The function block is called inside of a POU one time per task cycle without any conditions. This 

is called an invocation. 

2. After a TRUE status has been detected at the input xEnable, the internal state is switched from 

Dormant to Executing. 

3. The status of all inputs will not be sampled. They influence the current operation in every 

invocation (see CyclicAction). 

4. The output xBusy will be set to TRUE. 

5. The defined operation will be started (see the REPEAT-Loop inside CyclicAction). 

6. If the operating time for the current invocation is higher than the time (in µs) as specified by the 

input value xTimeLimit, the operation will be interrupted and continued in the next invocation 

(see xTimeLimit in CyclicAction). 

7. While working on the defined operation, a number of conditions can appear that lead to the exit 

from the Executing state. This means the value of the output variable xBusy will be set to FALSE 
and the internal state will be switched from Executing to one of the states Done or Error. This 

change will be mirrored to one of the output variables xDone or xError. Only one output 

variable of this set of variables can have the status TRUE at the same time. With the falling edge of 

xBusy the input variable xEnable is sampled and its inverted value is stored as a reset request 

(see xResetRequest in the method HandleErrorState). 

a. Ready Condition 2: 

If the operation has reached its ready condition without any error and timing constraints 

(see xComplete in CyclicAction), the output variable xDone is set to TRUE. This 

means the internal state is switched from Executing to Done. 

b. Error Condition: 

If an error condition was detected (see eErrorID in in CyclicAction), the output 

variable xError is set to TRUE. This means the internal state is switched from Executing 
to Error. In addition, one of the defined error codes (one value out of the local 

enumeration type ERROR) will be assigned to the output variable eErrorID. 

c. Timeout Condition: 

If the overall operating time of the defined operation is higher than the time (in µs) as 

specified by the input value udiTimeOut, a timeout condition will detected causing the 

output variable xError to be set to TRUE. This means the internal state is switched from 

Executing to Error. Furthermore, the output variable eErrorID is set to a special error 

code (ERROR.TIME_OUT). 

d. Abort Condition: 

If a status of FALSE was detected for the xEnable input variable, the abort condition is 

reached. This means the internal state is switched from Executing to Aborting. Any current 

action of the defined operation will be aborted. After this is done (see xComplete inside 

AbortAction), the internal state is switched from Aborting to Resetting. If an error 

condition was detected (see eErrorID in AbortAction), the output variable xError is 

set to TRUE. This means the internal state is switched from Aborting to Error (see error 

condition). 

8. As a reaction to the rising edge of the output variable xError it would be possible to set the 

status of the input variable xEnable again to TRUE (quick handshake). 

9. The value TRUE of one of the output variables xDone or xError must be stable for a minimum of 

one invocation. This means the internal states Done or Error must be active for a minimum of 

one invocation. This property guarantees that the values of the output variables are valid and 

stable for a minimum of one invocation. 

10. After a FALSE status for the input variable xEnable is detected (standard handshake) or a reset 

request is active (quick handshake), the internal state will be switched from Done or Error to 

Resetting. 

11. All outputs will be initialized to their default statuses (ResetAction). All claimed resources will 

be freed. Specifically, the output variables xDone and xError will be set to FALSE. After 

executing the code of the ResetAction the function block should be prepared for a new switch 

of the internal state from Dormant to Executing. 

                                                 
2 Sometimes a function block without an xDone output variable is required. In this case, please select the 

LConC or the LConTlC types. The state machine of this kind of function blocks will never switch to the 

Done state. An example of this is the MC_Power function block for motion control or the TCPServer 

function block as used in 3.7 Cooperation of various function blocks. 

12. After doing this reinitialization work, the internal state will be switched from Resetting to 

Dormant (see xComplete inside ResetAction). 

 

Example of an SFC diagram 

 

 
Figure 15: Example of a state diagram of LConTlTo in SFC 

An example of the implementation of LConTlTo in ST is shown in Appendix 1.4.4 LConTlTo. 

Appendix 1 Datasheets of the Edge Triggered and Level Controlled FBs 

In this chapter an overview of examples is provided, including source code, of possible implementations 

of both the edge as well as level controlled function blocks as part of the PLCopen Behaviour Model. The 

source code itself is available on the PLCopen website www.PLCopen.org in the same section and in 

different formats (with and without prefixes and with enum) and the files can be opened in any text editor 

and with a simple copy&paste added to your own environment. 

Appendix 1.1 General comments about the sample programs 
Some parts of the code for the different function blocks are very similar. Some parts are critical for a 

proper functionality according to the specification other parts are provides just as an example and should 

be adapted to the real requirements of a specific application. 

Methods marked with “PRIVATE FINAL” are a crucial part of the implementation. 

Methods marked with “PROTECTED” are a kind of template, they should be adapted to the real 

requirements of a specific application. 

Here is a list of the critical parts: 

 The order of the evaluation of the variables xExecute, xComplete, eErrorID and xAbort 

defines the behaviour in the case that these variables are set at the same time. A different order 

results in a different behaviour. 

 The handling of the variable xResetRequest defines the behaviour of the function block if a 

quick handshake operation is necessary. 

 The handling of the TimingController determines the behaviour of the function block in case 

an ERROR.TIME_OUT has to occur or at which exact point in time a specific function block will 

suspend the processing of its CyclicAction and return to its caller (xTimeLimit). 

 If an Abort Condition or an Error Condition was reached then it is not possible to return to a 

normal Ready Condition. An Abort Condition can be changed to an Error Condition. In an Error 

Condition only the eErrorID can be modified but not to the value ERROR.NO_ERROR. 

 An Error Condition that is not caused by a Timeout Condition (eErrorID ≠ ERROR.TIME_OUT) 

has the higher priority and must never be overwritten by ERROR.TIME_OUT. 

 Every state with a name ending with -ing like Executing or Resetting can run more than one 

invocation. In fact they will be executed as long as they need to reach their local Ready Condition, 

Abort Condition, Timeout Condition or Error Condition. 

Any state machine implementation of the PLCopen Behaviour Model has to consider these kinds of 

things to keep the conformance according to this document! 

In order to restrict any further interpretation as much as possible, in addition to the source code of each 

function block a set of timing diagrams for each function block was provided. 

 

  

http://www.plcopen.org/

Appendix 1.2 Overview of the functionalities 

  
 

  



PLCopen
®

 
for efficiency in automation 

 

Creating PLCopen Compliant Libraries  V 1.0 - Official Release © PLCopen (2017)  

Software Creation Guidelines May 4, 2017  page 42/89 

Appendix 1.3 Overview edge triggered FBs 

 
 

Note: In practice the types STATE and ERROR will be defined only once in a library and not in every FB 

as is done in the examples shown here. 

Appendix 1.3.1 ETrig 

ETrig (Edge Triggered | Not Abortable | Not Time Limited | Not Time Out Constraint) 

 

State Chart:

 
 

ETrig Implementation

ETrig Timing Diagram

Appendix 1.3.2 ETrigTl 

ETrigTl (Edge Triggered | Not Abortable | Time Limited | Not Time Out Constraint) 

 

ETrigTl State Chart

 
 

ETrigTl Implementation

ETrigTl Timing Diagram

Appendix 1.3.3 ETrigTo 

ETrigTo (Edge Triggered | Not Abortable | Not Time Limited | Time Out Constraint) 

 

ETrigTo State Chart

 
 

ETrigTo Implementation

ETrigTo Timing Diagram

Appendix 1.3.4 ETrigTlTo 

ETrigTlTo (Edge Triggered | Not Abortable | Time Limited | Time Out Constraint) 

 

ETrigTlTo State Chart

 
 

ETrigTlTo Implementation

ETrigTlTo Timing Diagram

Appendix 1.3.5 ETrigA 

ETrigA (Edge Triggered | Abortable | Not Time Limited | Not Time Out Constraint) 

ETrigA State Chart

 
 

ETrigA Implementation

ETrigA Timing Diagram

Appendix 1.3.6 ETrigATl 

ETrigATl (Edge Triggered | Abortable | Time Limited | Not Time Out Constraint) 

 

ETrigATl State Chart

 
 

ETrigATl Implementation

ETrigATl Timing Diagram

Appendix 1.3.7 EtrigATo 

ETrigATo (Edge Triggered | Abortable | Not Time Limited | Time Out Constraint) 

 

ETrigATo State Chart

 
 

ETrigATo Implementation

ETrigATo Timing Diagram

Appendix 1.3.8 ETrigATlTo 

ETrigATlTo (Edge Triggered | Abortable | Time Limited | Time Out Constraint) 

 

ETrigATlTo State Chart

 
 

ETrigATlTo Implementation

ETrigATlTo Timing Diagram

Appendix 1.4 Overview Level Controlled FBs 

Appendix 1.4.1 LCon 

LCon (Level Controlled | Not Time Limited | No Time Out Constraint | No Continuous Behaviour) 

 

LCon State Chart

 
 

LCon Implementation

Appendix 1.4.2 LConTl 

LConTl (Level Controlled | Time Limited | Not Time Out Constraint | No Continuous Behaviour) 

 

LConTl State Chart

 
 

LConTl Implementation

Appendix 1.4.3 LConTo 

LConTo (Level Controlled | Not Time Limited | Time Out Constraint | No Continuous Behaviour) 

 

LConTo State Chart

 
 

LConTo Implementation

Appendix 1.4.4 LConTlTo 

LConTlTo (Level Controlled | Time Limited | Time Out Constraint | No Continuous Behaviour) 

 

LConTlTo State Chart

 
 

LConTlTo Implementation

Appendix 1.4.5 LConC 

LConC (Level Controlled | Not Time Limited | Continuous Behaviour) 

 

LConC State Chart

 
 

LConC Implementation 

Appendix 1.4.6 LConTlC 

LConTlC (Level Controlled | Time Limited | Continuous Behaviour) 

 

LConTlC State Chart

 
 

LConTlC Implementation

Appendix 2 Example without using Object Oriented features 

It is not necessary to use object oriented features for creating function block libraries. With existing and 

classical environments this is very well possible and was always an approach in IEC 61131-3. 

For this reason an example of how to do that is shown here. 

If one can use the functionality of an edge triggered FB, one can define the basic ST code as listed in 

Chapter 5.1 The basic FB: ETrig, however without the object oriented features. This code could look like: 

 

The ETrigATlTo Function Block coded according to IEC 61131–3 2nd Edition  

(Exemplary Implementation)  

  
  

 FUNCTION_BLOCK ETrigATlTo  

  VAR_INPUT  

        // Rising edge starts defined operation  

        // FALSE ? resets the defined operation  

        // after ready condition was reached  

    xExecute: BOOL;  

        // command for abort the operation  

    xAbort: BOOL;  

        // max operating time per invocation  

        // [µs], 0 ? no operating time limit  

    udiTimeLimit: UDINT;  

        // max operating time per invocation  

        // [µs], 0 ? no operating time limit  

    udiTimeOut: UDINT;  

  END_VAR 

 

  VAR_OUTPUT  

        // ready condition reached  

    xDone: BOOL;  

        // operation is running  

    xBusy: BOOL;  

        // error condition reached  

    xError: BOOL;  

        // abort condition reached  

    xAborted : BOOL;  

        // error code describing error condition  

    eErrorID : ERROR;  

  END_VAR 

 

  VAR  

    tcTimingController : TimingController;  

    eState : STATE := STATE.DORMANT;  

    xFirstInvocation : BOOL := TRUE;  

    xAbortProposed : BOOL;  

    eErrorIDProposed : ERROR;  

    xResetRequest : BOOL;  

  END_VAR  

 

  VAR_TEMP  

    xAgain :BOOL;  

    xComplete : BOOL;  

    xTimeLimit : BOOL;  

    xTimeOut : BOOL;  

    xLocalAbort : BOOL;  

    eLocalErrorID : ERROR;  

  END_VAR  

  

  REPEAT  

     xAgain := FALSE;  

     CASE eState OF  

  STATE.DORMANT:  

       IF xExecute THEN  

                tcTimingController(xStartOperationTimer:=TRUE);  

                tcTimingController.xStartOperationTimer := FALSE;  

                xBusy := TRUE;  

                eState := STATE.STARTING;  

                xAgain := TRUE;  

           END_IF  

  

        STATE.STARTING:  

           IF NOT xAbort THEN (* StartAction *)  

              IF xFirstInvocation THEN  

                 // sample the input variables  

                 tcTimingController.udiTimeLimit := udiTimeLimit;  

                 tcTimingController.udiTimeOut := udiTimeOut;  

                 xFirstInvocation := FALSE;  

              END_IF  

  

                 // working to reach the locale ready condition  

                 // ? xComplete := TRUE  

                 // if an error condition is reached  

                 // ? set eLocalErrorID to a value other than  

 // ERROR.NO_ERROR  

 

                 xComplete := TRUE;  

                 eLocalErrorID := ERROR.NO_ERROR; 

  

            ELSE 

                 xAbortProposed := TRUE;  

            END_IF  

  

            tcTimingController(xTimeOut=>xTimeOut); 

 

            IF xTimeOut AND eLocalErrorID = ERROR.NO_ERROR THEN  

                 eLocalErrorID := ERROR.TIME_OUT;  

            END_IF  

  

            IF eLocalErrorID <> ERROR.NO_ERROR OR xAbortProposed THEN  

                 eState := STATE.CLEANING;  

                 xAgain := TRUE;  

            ELSIF xComplete THEN  

                 eState := STATE.EXECUTING;  

                 xAgain := TRUE;  

            END_IF  

  

        STATE.EXECUTING:  

            IF NOT (xAbort OR xAbortProposed) THEN  

                 tcTimingController(xStartInvocationTimer:=TRUE);  

                 tcTimingController.xStartInvocationTimer := FALSE;  

                 REPEAT (* CyclicAction *)  

                    // working to reach the ready condition  

                    // ? xComplete := TRUE  

                    // if the maximum invocation time is reached  

                    // ? xTimeLimit := TRUE  

                    // if the maximum operating time is reached  

                    // ? xTimeOut := TRUE  

                    // if an error condition is reached  

                    // ? set eLocalErrorID to a value other than  

    // ERROR.NO_ERROR 

 

                    tcTimingController(  

                       xTimeOut=>xTimeOut,  

                       xTimeLimit=>xTimeLimit  

                    );  

  

                    xComplete := TRUE;  

                    eLocalErrorID := ERROR.NO_ERROR;  

  

                  UNTIL xComplete OR  

                     xTimeOut OR xTimeLimit OR  

                     eLocalErrorID <> ERROR.NO_ERROR  

                  END_REPEAT  

 

              ELSE  

                  xAbortProposed := TRUE;  

              END_IF  

  

              tcTimingController(xTimeOut=>xTimeOut);  

              IF xTimeOut AND eLocalErrorID = ERROR.NO_ERROR THEN  

                  eLocalErrorID := ERROR.TIME_OUT;  

              END_IF  

  

              IF xComplete OR eLocalErrorID <> ERROR.NO_ERROR OR  

 xAbortProposed THEN  

                   eErrorIDProposed := eLocalErrorID;  

                   eState := STATE.CLEANING;  

                   xAgain := TRUE;  

              END_IF  

  

        STATE.CLEANING: (* CleanupAction *)  

            IF xAbortProposed THEN  

               // abort all running operations  

               // if an error condition is reached  

               // ? set eErrorID to a value other than ERROR.NO_ERROR  

               xLocalAbort := xAbortProposed;  

            END_IF  

  

            // if possible free as much allocated resources  

            // as possible  

            // working to reach the locale ready condition  

            // ? xComplete := TRUE  

            // if an error condition is reached  

            // ? set eLocalErrorID to a value other than ERROR.NO_ERROR  

            xComplete := TRUE;  

            eLocalErrorID := eErrorIDProposed;  

  

            IF xAbortProposed THEN  

                xComplete := FALSE;  

            ELSE  

                xLocalAbort := FALSE;  

            END_IF  

  

            IF eErrorIDProposed <> ERROR.NO_ERROR THEN  

                xComplete := FALSE;  

                xAbort := FALSE;  

            END_IF  

  

            IF eLocalErrorID <> ERROR.NO_ERROR THEN  

                eErrorIDProposed := eLocalErrorID;  

            END_IF  

  

            IF eLocalErrorID <> ERROR.NO_ERROR THEN  

                eState := STATE.ERROR;  

                xAgain := TRUE;  

            ELSIF xLocalAbort THEN  

                eState := STATE.ABORTED;  

                xAgain := TRUE;  

            ELSIF xComplete THEN  

                eState := STATE.DONE;  

                xAgain := TRUE;  

            END_IF  

  

       STATE.DONE:  

            IF xDone AND (xResetRequest OR NOT xExecute) THEN  

                eState := STATE.RESETTING;  

                xAgain := TRUE;  

            ELSE  

                xBusy := FALSE;  

                xDone := TRUE;  

                xResetRequest := NOT xExecute;  

                xAgain := FALSE; (* !!! *)  

            END_IF  

  

       STATE.ERROR:  

            IF xError AND (xResetRequest OR NOT xExecute) THEN  

                eState := STATE.RESETTING;  

                xAgain := TRUE;  

            ELSE  

                xBusy := FALSE;  

                xError := TRUE;  

                eErrorID := eErrorIDProposed;  

                xResetRequest := NOT xExecute;  

                xAgain := FALSE; (* !!! *)  

             END_IF  

  

        STATE.ABORTED:  

             IF xAborted AND (xResetRequest OR NOT xExecute) THEN  

                 eState := STATE.RESETTING;  

                 xAgain := TRUE;  

             ELSE  

                 xBusy := FALSE;  

                 xAborted := TRUE;  

                 xResetRequest := NOT xExecute;  

                 xAgain := FALSE; (* !!! *)  

             END_IF  

  

        STATE.RESETTING: (* ResetAction *)  

             // free all residual allocated resources  

             // reinitialize instance variables  

             // working to reach the locale ready condition  

             // ? xComplete := TRUE  

             xComplete := TRUE;  

  

             IF xComplete THEN  

                 xBusy := FALSE;  

                 xDone := FALSE;  

                 xError := FALSE;  

                 xAborted := FALSE;  

                 xAbortProposed := FALSE;  

                 eErrorIDProposed := ERROR.NO_ERROR;  

                 eErrorID := ERROR.NO_ERROR;  

                 eState := STATE.DORMANT;  

                 xAgain := xResetRequest; (* !!! *)  

                 xResetRequest := FALSE;  

                 xFirstInvocation := TRUE;  

             END_IF  

      END_CASE  

   UNTIL NOT xAgain  

   END_REPEAT; 

Appendix 3 Example of an intermediate interface 

In practice there are more than one naming convention standards. This can result in different layers of 

naming conventions: for the creation of the function block libraries, for the creation of the functional 

application, and for the mapping at the application level to the conventions of the user. 

These different levels can cooperate by encapsulating the functionalities on the different levels, or by 

using aliases, where one maps one name to another. 

One can call these intermediate interfaces, like an interface between the function block library and the 

application itself, both using different areas and so using different naming conventions while the 

application is building on top of the library. The different naming conventions can be linked (“alias”). 

 

One example is shown in the picture below. The public final function block MC_MoveVelocity on the 

left is based on the internal FB MotionCore on the right. By creating these two function blocks it is 

possible to encapsulate the complete implementation details. Using this technique the layout of the public 

function block is completely decoupled from the layout of the internal function block (see the different 

naming conventions for variable names). Because of the FINAL keyword it is not possible to extend this 

function block in another context. So no problems can occur after changing some implementation details 

(for example adding some local variables). Because of the INTERNAL keyword nobody can use a 

function block marked in this way outside of its defined context, its original library. This decouples these 

two layers. 

Appendix 4 Behaviour of inputs and outputs in PLCopen Motion 

Control FBs 

 

For the relevant inputs and outputs 

Input parameters With ‘Execute’ without ‘ContinuousUpdate’: The parameters are used with the 

rising edge of the ‘Execute’ input. To modify any parameter it is necessary to 

change the input parameter(s) and to trigger the ‘Execute’ input again. 

 

With ‘Execute’ combined with ‘ContinuousUpdate’: The parameters are used 

with the rising edge of the ‘Execute’ input. The parameters can be modified 

continuously as long as the ‘ContinuousUpdate’ is SET. 

 

With ‘Enable’:  The parameters are used with the rising edge of the enable input 

and can be modified continuously. 

Inputs exceeding 

application limits 

If a FB is commanded with parameters which result in a violation of application 

limits, the inputs are limited by the system or the instance of the FB generates 

an error. The consequences of this error for the axis are application specific and 

thus should be handled by the application program. 

Missing input 

parameters 

According to IEC 61131-3, if any parameter of a function block input is 

missing (“open”) then the value from the previous invocation of this instance 

will be used. In the first invocation the initial value is applied.  

Acceleration, 

Deceleration and 

Jerk inputs 

If the input ‘Deceleration’, ‘Acceleration’ or ‘Jerk’ is set to 0, the result is 

implementation dependent. There are several implementations possible, like 

one goes to the error state, one signals a warning (via a supplier specific 

output), one inhibits this in the editor, one takes the value as either specified in 

AxisRef or in the drive itself, or one takes a maximum value. Even if the 0 

value input is accepted by the system, please use with caution especially if 

compatibility is targeted. 

Output exclusivity With ‘Execute’: The outputs ‘Busy’, ‘Done’, ‘Error’, and ‘CommandAborted’ 

are mutually exclusive: only one of them can be TRUE on one FB. If ‘Execute’ 

is TRUE, one of these outputs has to be TRUE. 

Only one of the outputs ‘Active’, ‘Error’, ‘Done’ and ‘CommandAborted’ is set 

at the same time, except in MC_Stop where ‘Active’ and ‘Done’ can be set both 

at the same time 

 

With ‘Enable’: The outputs ‘Valid’ and ‘Error’ are mutually exclusive: only one 

of them can be TRUE on one FB. 

Output status With ‘Execute’: The ‘Done’, ‘Error’, ‘ErrorID’ and ‘CommandAborted’ outputs 

are reset with the falling edge of ‘Execute’ . However the falling edge of 

‘Execute’ does not stop or even influence the execution of the actual FB. It 

must be guaranteed that the corresponding outputs are set for at least one cycle 

if the situation occurs, even if execute was reset before the FB completed. 

If an instance of a FB receives a new execute before it finished (as a series of 

commands on the same instance), the FB won’t return any feedback, like 

‘Done’ or ‘CommandAborted’, for the previous action. 

 

With ‘Enable’: The ‘Valid’, ‘Enabled’, ‘Busy’, ‘Error’, and ‘ErrorID’ outputs 

are reset with the falling edge of ‘Enable’ as soon as possible. 

Behavior of Done 

output 

The ‘Done’ output is set when the commanded action has been completed 

successfully. 

With multiple Function Blocks working on the same axis in a sequence, the 

following applies: when one movement on an axis is interrupted with another 

movement on the same axis without having reached the final goal, ‘Done’ of 

the first FB will not be set. 

Behavior of Busy 

output 

With ‘Execute’: Every FB can have an output ‘Busy’, reflecting that the FB is 

not finished and new output values can be expected. ‘Busy’ is SET at the rising 

edge of ‘Execute’ and RESET when one of the outputs ‘Done’, ‘Aborted’, or 

‘Error’ is set. 

 

With ‘Enable’: Every FB can have an output ‘Busy’, reflecting that the FB is 

working and new output values can be expected. ‘Busy’ is SET at the rising 

edge of ‘Enable’ and stays SET as long as the FB is performing any action. 

 

It is recommended that the FB should be kept in the active loop of the 

application program for at least as long as ‘Busy’ is true, because the outputs 

may still change. 

Behavior of 

InVelocity, 

InGear, InTorque 

and InSync 

The outputs ‘InVelocity’, ‘InGear’, ‘InTorque’, and ‘InSync’ (from now on 

referred to as ‘Inxxx’) have a different behavior than the ‘Done’ output. 

As long as the FB is Active, ‘Inxxx’ is SET when the set value equals the 

commanded value, and will be RESET when at a later time they are unequal. 

For example, the InVelocity output is SET when the set velocity is equal to the 

commanded velocity. This is similar for ‘InGear’, ‘InTorque’, and ‘InSync’ 

outputs in the applicable FBs. 

‘Inxxx’ is updated even if ‘Execute’ is low as long as the FB has control of the 

axis (‘Active’ and ‘Busy’ are SET). 

The behavior of ‘Inxxx’ directly after ‘Execute’ is SET again while the 

condition of ‘Inxxx’ is already met, is implementation specific. 

 ‘Inxxx’ definition does not refer to the actual axis value, but must refer to the 

internal instantaneous setpoint. 

Output ‘Active’ The ‘Active’ output is required on buffered Function Blocks. This output is set 

at the moment the function block takes control of the motion of the according 

axis. For un-buffered mode the outputs ‘Active’ and ‘Busy’ can have the same 

value. 

For one axis, several Function Blocks might be busy, but only one can be active 

at a time. Exceptions are FBs that are intended to work in parallel, like 

MC_MoveSuperimposed and MC_Phasing’s, where more than one FB related 

to one axis can be active. 

Behavior of  

CommandAborted 

output 

‘CommandAborted’ is set, when a commanded motion is interrupted by another 

motion command. 

The reset-behavior of ‘CommandAborted’ is like that of ‘Done’. When 

‘CommandAborted’ occurs, the other output-signals such as ‘InVelocity’ are 

reset. 

Enable and Valid The ‘Enable’ input is coupled to a ‘Valid’ output. ‘Enable’ is level sensitive and 

‘Valid’ shows that a valid set of outputs is available at the FB. 

The ‘Valid’ output is TRUE as long as a valid output value is available and the 

‘Enable’ input is TRUE. The relevant output value can be refreshed as long as 

the input ‘Enable’ is TRUE. 

If there is a FB error, the output is not valid (‘Valid’ set to FALSE). When the 

error condition disappears, the values will reappear and ‘Valid’ output will be 

set again. 

Position versus 

distance 

‘Position’ is a value defined within a coordinate system. ‘Distance’ is a relative 

measure related to technical units. ‘Distance’ is the difference between two 

positions. 

Sign rules The ‘Acceleration’, ‘Deceleration’ and ‘Jerk’ are always positive values. 

‘Velocity’, ‘Position’ and ‘Distance’ can be both positive and negative. 

Error Handling  

Behavior 

All blocks can have two outputs, which deal with errors that can occur while 

executing that Function Block. These outputs are defined as follow: 

Error  Rising edge of ‘Error’ informs that an error occurred during the 

execution of   the Function Block. 

ErrorID Error identification (Extended parameter) 

‘Done’, ‘InVelocity’, ‘InGear’, ‘InTorque’, and ‘InSync’ mean successful 

completion so these signals are logically exclusive to ‘Error’. 

Types of errors: 

 Function Blocks (e.g. parameters out of range, state machine violation 

attempted) 

 Communication 

 Drive 

Instance errors do not always result in an axis error (bringing the axis to 

‘ErrorStop’) 

The error outputs of the relevant FB are reset with falling edge of ‘Execute’ and 

‘Enable’. The error outputs at FBs with ‘Enable’ can be reset during operation 

(without a reset of ‘Enable’). 

FB Naming In case of multiple libraries within one system (to support multiple drive / 

motion control systems), the FB naming may be changed to 

“MC_FBname_SupplierID”. 
Naming conventions 

ENUM types 
Due to the naming constraints in the IEC standard on the uniqueness of variable 

names, the ‘mc’ reference to the PLCopen Motion Control namespace is used 

for the ENUMs. 

In this way we avoid the conflict that using the ENUM types ‘positive’ and 

‘negative’ for instance with variables with these names throughout the rest of 

the project since they are called mcPositive and mcNegative respectively. 

Table 1: General Rules 

 

The behavior of the ‘Execute’ / ‘Done’ style FBs is as follows: 

Figure 16: The behavior of the ‘Execute’ / ‘Done’ in relevant FBs 

 

The behaviour of the ‘Execute’ / ‘Inxx’ style FBs is as follows: 

 

Execute

Busy

InXxx

Error

CommandAborted

Case 1 Case 2 Case 3

Active

 
Figure 17: The behavior of the ‘Execute’ / ‘Inxx’ in relevant FBs 

